

Summary

This application note shows how to configure the XC9572XL CPLD on the XSA Board so its parallel port interface emulates the functions of the Xilinx Parallel Cable III. This lets you use the Xilinx WebPACK JTAG tools with the XSA Board through its simple 25-wire downloading cable.

Why Emulate the Parallel Cable III?

Xilinx WebPACK software contains tools for downloading and testing SpartanII FPGAs through their JTAG interface. One of the ways these tools access the FPGA is through a Parallel Cable III connected from the FPGA to the parallel port of a PC. A schematic for the Parallel Cable III (henceforth referred to as PCBLIII) is shown in Figure 1.

The SpartanII FPGA on the XSA Board is accessed from the PC parallel port through a simple 25-wire cable that connects to an XC9572XL CPLD on the XSA Board. The CPLD can be programmed to pass signals from the parallel port to and from the JTAG pins of the SpartanII device. The XSA Board is supplied with a default CPLD configuration that lets you download bitstreams to the SpartanII using the GXSLOAD utility provided by XESS.

This application note describes an alternate circuit that allows the XSA Board CPLD to emulate the JTAG functions of the PCBLIII. By loading this circuit into the CPLD, you can use all the Xilinx downloading and testing tools with the XSA Board through the simple downloading cable provided by XESS.

VHDL for the Parallel Cable III Emulator

Listing 1 shows the VHDL code for the PCBLIII emulator that is programmed into the XC9572XL CPLD on the XSA Board. This interface provides two functions:

- It transfers configuration bitstreams from the PC to the SpartanII FPGA using the JTAG interface.

- After the SpartanII FPGA is configured and its DONE pin goes high, the JTAG interface is used to readback and/or test the FPGA.

How the VHDL implements these functions is described below.

Line 39 disables the Flash RAM until after the FPGA is configured. The clock signal from the programmable oscillator is passed to a dedicated clock pin of the SpartanII on line 42.

Line 45 outputs a high logic level to status pin S3 of the parallel port. The Xilinx software checks for a high level on this status pin which indicates that power is being supplied to the PCBLIII.

The Xilinx software also checks for the presence of the PCBLIII by looping a signal from parallel port data pin D6 back through two of the status pins S5 and S7. Line 50 handles the loop from D6 back to S5. The CPLD cannot pass D6 to S7, however, because its TDO pin is already attached to S7. Therefore, the shunt on jumper J9 has to be moved to the **xi** position to manually connect D6 and S7.

Line 52 drives the mode pins of the SpartanII FPGA to set it in the slave-serial configuration mode. This doesn't really do anything since the SpartanII will be programmed through its JTAG interface. Line 53 uses a pullup on the CPLD pin to hold the /PROGRAM pin of the SpartanII at a high logic level, thus preventing accidental erasure of the FPGA configuration.

Lines 57–59 connect parallel port data pins D2, D1, and D0 to the SpartanII TMS, TCK, and TDI JTAG pins if D3 is low. The TMS, TCK and TDI inputs are allowed to float when D3 is high.

Line 62 passes the FPGA JTAG TDO signal back to the PC through status pin S4 of the parallel port. S4 is driven low when data pin D4 is low.

The PCBLIII emulator I/O pin assignments for the CPLD on the XSA Board are shown in Listing 2.

Using the Parallel Cable III Emulator

First, connect the XSA Board to the parallel port of a PC through the simple 25-wire cable provided by XESS. Make sure the shunt on jumper J9 is in the **xs** position. Then download the piiijtag.svf file into the XC9572XL CPLD using the GXSLOAD tool from XESS. Now move the shunt on J9 to the **xi** position. At this point, the bitstream downloading portion of the PCBLIII emulator is active.

Next, double-click the Configure Device (iMPACT) icon in the Process pane of the WebPACK **Project Navigator** window. (This discussion assumes you already have a SpartanII design synthesized and implemented in WebPACK.) The **iMPACT** window will appear. Within a few seconds the XSA Board will be probed and the JTAG chain consisting of a single XC2S100 FPGA will be detected.

In the lower pane of the **iMPACT** window, click on the xc2s100 object to select this device as the target for configuration. Then select the Operations→Program... menu item and the **Program Options** window will appear. Click on the OK button and the bitstream for your design will download into the SpartanII FPGA on the XSA Board.




Figure 1: Xilinx Parallel Cable III schematic. The line numbers of the VHDL code in Listing 1 associated with each schematic element are shown.

Listing 1: VHDL code for the Parallel Cable III emulator.

```
1  library ieee;
2  use ieee.std_logic_1164.all;
3
4  entity piiv is
5      port(
6          -- parallel port data and status pins
7          ppd:      in std_logic_vector(6 downto 0);
8          pps:      out std_logic_vector(5 downto 3);
9
10         -- programmable oscillator
11         clk: in std_logic;
12
13         -- Spartan2 FPGA pins
14         S2_tck:      out std_logic; -- driver to Spartan2 JTAG clock
15         S2_tms:      out std_logic; -- driver to Spartan2 JTAG mode input
16         S2_tdi:      out std_logic; -- driver to Spartan2 JTAG serial data input
17         S2_tdo:      in std_logic; -- input from Spartan2 JTAG serial data output
18         S2_cclk:      out std_logic; -- driver to Spartan2 config clock
19         S2_prog_n: out std_logic; -- driver to Spartan2 /PROGRAM pin
20         S2_done:      in std_logic; -- input from Spartan2 DONE pin
21         S2_din:      out std_logic; -- driver to Spartan2 config serial data input
22         S2_m:         out std_logic_vector(0 downto 0); -- Spartan2 config mode pins
23         S2_clk:       out std_logic; -- clock output to Spartan2
24
25         fce_n:        out std_logic -- Flash chip-enable
26     );
27 end piiv;
28
29 architecture arch of piiv is
30     constant NO: std_logic := '0';
31     constant YES: std_logic := '1';
32     constant LO: std_logic := '0';
33     constant HI: std_logic := '1';
34     constant SLAVE_SERIAL_MODE: std_logic_vector(0 downto 0) := "1";
35     component pullup port(O: out std_logic); end component;
36 begin
37
38     -- disable Flash chip when Spartan2 is not configured
39     fce_n      <= HI when S2_done=NO else 'Z';
40
41     -- send the programmable oscillator clock to the Spartan2
42     S2_clk     <= clk;
43
44     -- the XSA power status is sent back through the parallel port status pin 3
45     pps(3)     <= HI; -- tell the PC that the VCC for the XSA board is OK
46     -- the cable is detected by sending data through data pin 6 and returning
47     -- it on status pins 5 and 7. Status pin 7 is used by the JTAG TDO
48     -- pin of the XC9500 CPLD on the XSA Board, so place a shunt at position "xi"
49     -- of jumper J9 to make this connection.
50     pps(5)     <= ppd(6);
51
52     S2_m       <= SLAVE_SERIAL_MODE; -- set Spartan2 config mode pins
53     u0: pullup port map(O=>S2_prog_n); -- place a pullup on the Spartan2 PROGRAM pin
54
55     -- drive the Spartan2 JTAG pins from the parallel port when tristate
56     -- control pin (parallel port data pin 3) is low.
57     S2_tms     <= ppd(2) when ppd(3)=LO else 'Z';
58     S2_tck     <= ppd(1) when ppd(3)=LO else 'Z';
59     S2_tdi     <= ppd(0) when ppd(3)=LO else 'Z';
60
61     -- the JTAG TDO output is sent back through the status pin
62     pps(4)     <= S2_tdo when ppd(4)=HI else LO;
```

63 `end arch;`

Listing 2: User-constraint file for CPLD pin assignments.

```
1  #
2  # pin assignments for the XC9572XL CPLD chip on the XSA Board
3  #
4
5  # Spartan2 FPGA connections to CPLD
6  net S2_clk      loc=p42;
7  net S2_tck      loc=p13;
8  # net S2_dout    loc=p18;
9  net S2_tms      loc=p18;
10 # net S2_din     loc=p2;
11 # net S2_wr_n    loc=p19;
12 net S2_tdo      loc=p19;
13 # net S2_cs_n    loc=p15;
14 net S2_tdi      loc=p15;
15 # net S2_init_n  loc=p38;
16 net S2_done     loc=p40;
17 net S2_prog_n   loc=p39;
18 # net S2_cclk    loc=p16;
19 net S2_m<0>     loc=p36;
20 # net S2_d<0>    loc=p2;
21 # net S2_d<1>    loc=p4;
22 # net S2_d<2>    loc=p5;
23 # net S2_d<3>    loc=p6;
24 # net S2_d<4>    loc=p7;
25 # net S2_d<5>    loc=p8;
26 # net S2_d<6>    loc=p9;
27 # net S2_d<7>    loc=p10;
28
29 # Flash RAM
30 # net fd<0>      loc=p2;
31 # net fd<1>      loc=p4;
32 # net fd<2>      loc=p5;
33 # net fd<3>      loc=p6;
34 # net fd<4>      loc=p7;
35 # net fd<5>      loc=p8;
36 # net fd<6>      loc=p9;
37 # net fd<7>      loc=p10;
38 # net fa<0>      loc=p1;
39 # net fa<1>      loc=p64;
40 # net fa<2>      loc=p63;
41 # net fa<3>      loc=p62;
42 # net fa<4>      loc=p61;
43 # net fa<5>      loc=p60;
44 # net fa<6>      loc=p59;
45 # net fa<7>      loc=p58;
46 # net fa<8>      loc=p45;
47 # net fa<9>      loc=p44;
48 # net fa<10>     loc=p57;
49 # net fa<11>     loc=p43;
50 # net fa<12>     loc=p56;
51 # net fa<13>     loc=p46;
52 # net fa<14>     loc=p47;
53 # net fa<15>     loc=p52;
54 # net fa<16>     loc=p51;
55 # net fa<17>     loc=p48;
56 # net frst_n     loc=p50;# Flash reset
57 # net foe_n      loc=p12;# Flash output-enable
58 # net fwe_n      loc=p49;# Flash write-enable
59 net fce_n       loc=p11;# Flash chip-enable
60
61 # DIP and pushbutton switches
62 # net dipsw<1>   loc=p47;
```

```
63 # net dipsw<2>      loc=p52;
64 # net dipsw<3>      loc=p51;
65 # net dipsw<4>      loc=p48;
66
67 # 7-segment LEDs
68 # net s<0>          loc=p10;
69 # net s<1>          loc=p2;
70 # net s<2>          loc=p9;
71 # net s<3>          loc=p8;
72 # net s<4>          loc=p5;
73 # net s<5>          loc=p7;
74 # net s<6>          loc=p6;
75 # net dp           loc=p4;
76
77 # programmable oscillator
78 net clk            loc=p17;
79
80 # parallel port
81 net ppd<0>          loc=p33;
82 net ppd<1>          loc=p32;
83 net ppd<2>          loc=p31;
84 net ppd<3>          loc=p27;
85 net ppd<4>          loc=p25;
86 # net ppd<5>          loc=p24;
87 net ppd<6>          loc=p23;
88 # net ppd<7>          loc=p22;
89 net pps<3>          loc=p34;
90 net pps<4>          loc=p20;
91 net pps<5>          loc=p35;
92
```
